
*** COUNTY-BASED MAPS OF POPULATION AND PERCENT OF POPULATION AGE 65+;*** COUNTY-BASED MAPS OF POPULATION AND PERCENT OF POPULATION AGE 65+;

*** reset all graphics options

goptions reset=all;

*** location of census data sets;

libname cc 'i:\census\nj\sf1\v8';

*** location of map data sets created from census shapefiles;

libname m2 'i:\sas\maps2000';

*** 2000 census populations from summary file 1 - table pct12;

data njcou (keep=county pop pct65);
set cc.pct12;
where sumlev eq '050';
pct65  = 100 * sum(of pct012068-pct012105 pct012172-pct012209) / pct012001;
pct65  = round(pct65,.1);
pop    = pct012001;
run;

*** place population and percent population 65+ in quartiles - use proc rank;

proc rank data=njcou out=njcou_ groups=4;
var pop pct65;
ranks r_pop r_pct65;
run;

*** create formats to use with maps based on quartiles;

*** total population;

proc sort data=njcou_;
by pop;
run;

data fmt;
retain fmtname "pop4_" low;
set njcou_;
by r_pop;
if first.r_pop then low=pop;
if last.r_pop then do;
   start = r_pop;
   label = put(low,comma10.) || "-" || left(put(pop,comma10.));
   output;
end;
run;

proc format cntlin=fmt;
run;



NEW JERSEY MAPS - 2000 CENSUS DATA 2

*** percent population age 65+;

proc sort data=njcou_;
by pct65;
run;

data fmt;
retain fmtname "pct65_" low;
set njcou_;
by r_pct65;
if first.r_pct65 then low=pct65;
if last.r_pct65 then do;
   start = r_pct65;
   label = put(low,10.1) || "%-" || trim(left(put(pct65,10.1))) || '%';
   output;
end;
run;

proc format cntlin=fmt;
run;

*** use map created from us census website shapefile;
*** converted to map data set with proc mapimport (version 9);

*** map is unprojected - project the map;

proc gproject data=m2.njcounties out=cou asis;
id county;
run;

*** create a map;

proc gmap
map=cou
data=njcou_
;
id county;
choro r_pop;
format r_pop pop4_.;
run;
quit;

*** use map from sas-supplied map data sets;
*** select NEW JERSEY counties (FIPS state number 34);

proc gproject data=maps.counties out=cou asis;
where state eq 34;
id county;
run;

*** county is numeric in sas-supplied map data set, character in census data;
*** convert county in map data set to character;

data cou;
set cou (rename=(county=temp));
county = put(temp,z3.);
drop temp;
run;

*** add a titles;

title1 'NEW JERSEY COUNTIES';
title2 '2000 CENSUS POPULATION';

*** create a map;

proc gmap
map=cou
data=njcou_
;
id county;
choro r_pop;
format r_pop pop4_.;
run;
quit;

*** choose some new patterns (sequential reds from color brewer web site);
*** http://www.personal.psu.edu/faculty/c/a/cab38/ColorBrewerBeta.html ;

pattern1 v=s c='cxfee5d9';
pattern2 v=s c='cxfcae91';
pattern3 v=s c='cxfb6a4a';
pattern4 v=s c='cxcd181d';

*** create a map;

proc gmap
map=cou
data=njcou_
;
id county;
choro r_pop;
format r_pop pop4_.;
run;
quit;

*** reset all graphics options, select font...;

goptions
reset=all
ftext='Arial/bo'
htext=3
gunit=pct
cback=graydd
;

pattern1 v=s c='cxfee5d9';
pattern2 v=s c='cxfcae91';
pattern3 v=s c='cxfb6a4a';
pattern4 v=s c='cxcd181d';

*** move legend...;

legend1
mode=share
origin=(5,50)
shape=bar(2,3)
across=1
label=none
;

*** create a map;
*** outline counties, use the new legend, add a note;

proc gmap
map=cou
data=njcou_
;
id county;
choro r_pop /
coutline=gray33
legend=legend1
;
format r_pop pop4_.;
note h=4 'NEW JERSEY COUNTIES' j=l '2000 CENSUS POPULATION';
run;
quit;

*** create a gif file (specify GIF device driver);

goptions
reset=all
dev=gif
gsfname=gout
ftext='Arial/bo'
htext=3
gunit=pct
cback=graydd
;

pattern1 v=s c='cxfee5d9';
pattern2 v=s c='cxfcae91';
pattern3 v=s c='cxfb6a4a';
pattern4 v=s c='cxcd181d';

*** move legend...;

legend1
mode=share
origin=(5,50)
shape=bar(2,3)
across=1
label=none
;

*** destination for GIF file;

filename gout 'i:\njcou_2000.gif';

*** create a map;

proc gmap
map=cou
data=njcou_
;
id county;
choro r_pop /
coutline=gray33
legend=legend1
;
format r_pop pop4_.;
note h=4 'NEW JERSEY COUNTIES' j=l '2000 CENSUS POPULATION';
run;
quit;

*** create a pdf file (no device specified);

goptions
reset=all
ftext='helvetica/bo'
htext=1.75
gunit=pct
cback=graydd
;

pattern1 v=s c='cxfee5d9';
pattern2 v=s c='cxfcae91';
pattern3 v=s c='cxfb6a4a';
pattern4 v=s c='cxcd181d';

legend1
mode=share
origin=(5,50)
shape=bar(3,2)
across=1
label=none
;

*** use ODS to create a PDF file;

ods listing close;
ods pdf file='i:\njcou_2000.pdf' notoc;

proc gmap
map=cou
data=njcou_
;
id county;
choro r_pop /
coutline=gray33
legend=legend1
;
format r_pop pop4_.;
note 
h=2.5 'NEW JERSEY COUNTIES' ls=1
j=l '2000 CENSUS POPULATION' ls=1;
run;
choro r_pct65 /
coutline=gray33
legend=legend1
;
format r_pct65 pct65_.;
note h=2.5 'POPULATION 65+' ls=1;
run;
quit;

ods pdf close;
ods listing;

*** CENSUS TRACT-BASED MAPS OF POPULATION AND PERCENT OF POPULATION AGE 65+;*** CENSUS TRACT-BASED MAPS OF POPULATION AND PERCENT OF POPULATION AGE 65+; 

*** statements similar to those used in county-based map SAS code;

libname cc 'i:\census\nj\sf1\v8';

libname m2 'i:\sas\maps2000';

data njtr (keep=county tract pop pct65);
set cc.pct12;
where sumlev eq '140';
pct65 = .;
if pct012001 gt 0 then 
pct65 = 100 * sum(of pct012068-pct012105 pct012172-pct012209) / pct012001;
pct65 = round(pct65,.1);
pop   = pct012001;
run;

proc rank data=njtr out=njtr_ groups=5;
var pop pct65;
ranks r_pop r_pct65;
run;

proc sort data=njtr_;
by pop;
run;

data fmt;
retain fmtname "pop5_" low;
set njtr_;
by r_pop;
if first.r_pop then low=pop;
if last.r_pop then do;
   start = r_pop;
   label = put(low,comma10.) || "-" || left(put(pop,comma10.));
   output;
end;
run;

proc format cntlin=fmt;
run;

proc sort data=njtr_;
by pct65;
run;

data fmt;
retain fmtname "pct65_" low;
set njtr_;
by r_pct65;
if first.r_pct65 then low=pct65;
if last.r_pct65 then do;
   start = r_pct65;
   label = put(low,10.1) || "%-" || trim(left(put(pct65,10.1))) || '%';
   output;
end;
run;

proc format cntlin=fmt;
run;

*** use map created from us census website shapefile;
*** converted to map data set with proc mapimport (version 9);

*** map is unprojected - project the map;

proc gproject data=m2.njtracts out=njctr asis;
id county tract;
run;

*** tract numbers in census data are zero-padded on right;
*** fix tract numbers in map data set (change NULLS to ZEROES);

data njctr;
set njctr;
tract = translate(tract,'0','00'x);
run;

goptions
reset=all
ftext='helvetica/bo'
htext=1.75
gunit=pct
cback=cyan
;         

pattern1 v=s c='cxf0f9e8';
pattern2 v=s c='cxbae4bc';
pattern3 v=s c='cx7bccc4';
pattern4 v=s c='cx43a2ca';
pattern5 v=s c='cx0868ac';

legend1
mode=share
origin=(5,50)
shape=bar(3,2)
across=1
label=none
;            

ods listing close;
ods pdf file='i:\njtr.pdf' notoc;

*** create a map --- no outlines on tracts;

proc gmap
map=njctr
data=njtr_
;
id county tract;
choro r_pop/
legend=legend1
;
format r_pop pop5_.;
note
h=2.25 'NEW JERSEY CENSUS TRACTS' ls=1
j=l '2000 CENSUS POPULATION' ls=1;       
run;
choro r_pct65/
legend=legend1
;
format r_pct65 pct65_.;
note h=2.25 'POPULATION 65+' ls=1;   
run;
quit;

ods pdf close;
ods listing;

*** BERGEN COUNTY;*** BERGEN COUNTY;
*** CENSUS TRACT-BASED MAPS OF POPULATION AND PERCENT OF POPULATION AGE 65+;*** CENSUS TRACT-BASED MAPS OF POPULATION AND PERCENT OF POPULATION AGE 65+;

libname cc 'i:\census\nj\sf1\v8';

libname m2 'c:\sas\maps2000';

*** select Bergen County from census data;

data bergentr (keep=tract pop pct65);
set cc.pct12;
where sumlev eq '140' and county eq '003';
pct65 = 100 * sum(of pct012068-pct012105 pct012172-pct012209) / pct012001;
pct65 = round(pct65,.1);
pop   = pct012001;
run;

proc rank data=bergentr out=bergentr_ groups=5;
var pop pct65;
ranks r_pop r_pct65;
run;

proc sort data=bergentr_;
by pop;
run;

data fmt;
retain fmtname "pop5_" low;
set bergentr_;
by r_pop;
if first.r_pop then low=pop;
if last.r_pop then do;
   start = r_pop;
   label = put(low,comma10.) || "-" || left(put(pop,comma10.));
   output;
end;
run;

proc format cntlin=fmt;
run;

proc sort data=bergentr_;
by pct65;
run;

data fmt;
retain fmtname "pct65_" low;
set bergentr_;
by r_pct65;
if first.r_pct65 then low=pct65;
if last.r_pct65 then do;
   start = r_pct65;
   label = put(low,10.1) || "%-" || trim(left(put(pct65,10.1))) || '%';
   output;
end;
run;

proc format cntlin=fmt;
run;



NEW JERSEY MAPS - 2000 CENSUS DATA 9

*** select Bergen County from map data set;

proc gproject data=m2.njtracts out=ctr asis;
where county eq '003';
id tract;
run;

data ctr;
set ctr;
tract = translate(tract,'0','00'x);
run;

*** create maps in a PDF file, change orientation to landscape;

options orientation=landscape;

goptions
reset=all
ftext='helvetica/bo'
htext=1.75
gunit=pct
cback=cyan
rotate=landscape
border
;

pattern1 v=s c='cxf0f9e8';
pattern2 v=s c='cxbae4bc';
pattern3 v=s c='cx7bccc4';
pattern4 v=s c='cx43a2ca';
pattern5 v=s c='cx0868ac';

legend1
mode=share
origin=(5,30)
shape=bar(3,2)
across=1
label=none
;

*** two maps on one page --- specify startpage option;

ods listing close;
ods pdf file='z:\bergentr_2000.pdf' notoc startpage=never;

*** choose size for maps;

goptions 
hsize=5.2 in
vsize=7.9 in
;

proc gmap
map=ctr
data=bergentr_
all
;
id tract;
choro r_pop/
legend=legend1
coutline=cyan
;
format r_pop pop5_.;
note h=2.5
j=r 'BERGEN COUNTY CENSUS TRACTS ' ls=1
j=r '2000 CENSUS POPULATION ' ls=1;
run;

*** move to center of page for second map;

goptions horigin=5.5 in;

choro r_pct65/
legend=legend1
coutline=cyan
;
format r_pct65 pct65_.;
note h=2.5 j=r 'PERCENT 65+ ' ls=1;
run;
quit;

ods pdf close;
ods listing;

Mike Zdeb
U@Albany School of Public Health

msz03@health.state.ny.us

518-402-6479

examples from chapter 5 of "Maps Made Easy Using SAS"...

http://www.albany.edu/~msz03/maps.html

SAS web site mapping...

http://support.sas.com/rnd/datavisualization/mapsonline/html/

census bureau shapefiles (use with PROC MAPIMPORT to create map data sets)...

http://www.census.gov/geo/www/cob/bdy_files.html

Color Brewer (for selecting map colors)...

http://www.personal.psu.edu/faculty/c/a/cab38/ColorBrewerBeta.html

or...

http://www.colorbrewer.org

buy a book...(shameless self-promotion)...$24.95 (cheap)...

http://www.sas.com/apps/pubscat/bookdetails.jsp?catid=1&pc=57495

The world is yours with Maps Made Easy Using SAS.  Bring your data tolife and add meaning to your information with Maps
Made Easy Using SAS.  Abundant real-world examples and a tutorial approach help new users create maps easily and quickly. 
You will learn the basic mapping components, including map and response data sets as well as simpleSAS/GRAPH statements. 
With in-depth examples you will move on to more advanced mapping techniques, such as annotating maps and producing
customized maps and output.  The process used to annotate maps is demystified and described in clear, easy-to-follow steps. 
You will produce data-driven, updatable maps in GIF format for use in Web-based presentations and other applications.  Also
presented are details on creating more complicated choropleth maps.  These include maps that combine geographic areas with
internal boundaries removed, maps that display multiple geographic areas, and clipped maps.  Enhance your data
presentations with this well-organized guide.

